

\$SIN10.INC

```
-----
RSEG PAGE0
LocSym +
-----
```

PUBLIC MEMODAT,PREP_H,VISU_H,INIHEU,MINUTE

EXTERN LPRO,BCD_HEX,HEX_ASC,SAUV_B,REMET_B,TXTHEUR,FinPowerUp,MINUT
EXTERN ASP,RSP,DIFAFF,CAS_SEC,DEMARRE,P_DIFAFF,KP_ADRE,DemandeDiffusion
EXTERN TABJOU

```
*****
;==* 1'heure est calculee toutes les secondes
;== le jours de la semaine et le quantieme sont systematiquement recalculés
;
; 1'heure est rangees en 0160H
; 0160H= secondes en bcd
; 0161H= minutes en bcd
; 0162H= heures en bcd
; 0163H= jours en bcd
; 0164H= mois en bcd
; 0165H= annee de 0 a 3
; 0166H= siecles en bcd (ex 19)
; 0167H= annee courante (ex 95) en bcd
; 0168H= centaines du quantieme
; 0169H= diz et unit du quantieme en bcd
; 016AH= jour de la semaine de 0 a 6
; 016BH= jour de la semaine en binaire x D S V J M M L

;*****
; calcul de 1'heure
```

FINDAT: LJMP FINDAT2

MEMODAT:MOV R9L,#0
MOV R9H,#0
MOV DPTR,#CAS_SEC ;pointe la case de la seconde
CALL INC_BCD ;incrimente en bcd la case
CJNE A,#60H,PLUS1

PLUS1: JC FINDAT
SETB MINUT
CLR A
MOVX @DPTR,A

;incrimente les minutes
INC DPL
CALL INC_BCD
CJNE A,#60H,PLUS2

PLUS2: JC FINDAT
CLR A
MOVX @DPTR,A

;incrimente les heures
INC DPL
CALL INC_BCD
CJNE A,#24H,PLUS3

PLUS3: JC FINDAT
CLR A
MOVX @DPTR,A

;incrimente les jours du mois
INC DPL
CALL INC_BCD
INC DPL ;pointe le mois
MOVX A,@DPTR
MOV R8L,A
INC DPL
MOVX A,@DPTR
MOV R8H,A ;pointe 1'annee de 0 a 3 0=annee bissextile
DEC DPL
DEC DPL

; recup le nombre de jours maxi du mois
PUSH DPH
PUSH DPL
MOV DPTR,#TABJ1 ;table des jours avec annee bisse
JZ TAB2
MOV DPTR,#TABJ2
TAB2: MOV A,R8L ;recup le mois
CALL BCD_HEX
DEC A

```

        MOVC A,@A+DPTR
        MOV B,A
        POP DPL
        POP DPH          ;dptr pointe le jour du mois
        MOVX A,@DPTR      ;jours du mois +1
        CJNE A,B,FINDAT2  ;teste si dernier jour du mois
        MOV A,#1          ;jour = 1
        MOVX @DPTR,A
;incremente le mois
        INC DPL
        CALL INC_BCD
        CJNE A,#13H,FINDAT2
        MOV A,#1
        MOVX @DPTR,A
; incremente l'annee de 0 a 3
        INC DPL
        MOVX A,@DPTR
        INC A
        MOV B,A
        MOVX @DPTR,A
        CJNE A,#4,SIMIN
SIMIN:  JC ANNEE
        CLR A
        MOV B,A
        MOVX @DPTR,A
;incremente l'annee
ANNEE:  INC DPL
        INC DPL
        CALL INC_BCD
        JNC FINDAT2
;incremente les siecles
        DEC DPL
        CALL INC_BCD

;===== calcul du quantieme, du jour de la semaine, et de l'annee de 0 a 3
FINDAT2:CALL JSEMAI
        MOV DPTR,#CAS_SEC +0AH
        MOVX @DPTR,A      ;jsem de 0 a 6
        INC DPL           ;j sem code en binaire

; conversion de 0a6 en binaire
        JNZ DIRECT
        MOV A,#7
DIRECT: MOV R8L,A
        SETB C
        CLR A
LOOPS:  RLC A
        DJNZ R8L,LOOPS
        MOVX @DPTR,A

; calcul de l'annee de 0 a 3
        MOV DPTR,#CAS_SEC + 7
        MOVX A,@DPTR      ;recup l'annee en cours
        CALL BCD_HEX
        MOV B,#4
        DIV AB
        DEC DPL
        DEC DPL
        MOV A,B
        MOVX @DPTR,A
        RET

;***** fin du calcul de l'heure
;test si lancement des procedures de la minute
MINUTE: JNB MINUT,NOMINU
        JNB DEMARRE,NORMAL
        CLR DEMARRE
        SETB FinPowerUp

;===lancement des procedures de la minute
NORMAL: MOV R9L,#8
        MOV R9H,#0F0H
PROSU:  MOV A,R9H
        CALL LPRO
        INC R9H
        INC R9H
        DJNZ R9L,PROSU
NOMINU: RET

TABJ1:  DB 32H,30H,32H,31H,32H,31H,32H,32H,31H,32H,31H,32H

```

```

TABJ2:  DB 32H,29H,32H,31H,32H,31H,32H,32H,31H,32H,31H,32H

INC_BCD:MOVX A,@DPTR
        ADD A,#1
        DA A
        MOVX @DPTR,A
        RET

;***** calcul du jour de la semaine et du quantieme

JSEMAI: MOV DPTR,#CAS_SEC + 3 ;POINTE LES JOURS du mois
        MOVX A,@DPTR
        MOV R8L,A ;R8L=JOURS DU MOIS
        INC DPL
        MOVX A,@DPTR
        MOV R8H,A ;R8H=MOIS
        INC DPL
        INC DPL
        INC DPL
        MOVX A,@DPTR
        MOV R9L,A ;R9L=ANNEE EN COURS

        MOV R9H,#89H ;ANNEE REFERENCE POUR LE CALCUL
        MOV REL,#00H ;JOUR DE LA SEMAINE DU 1/1/1989

DEBDEC: MOV A,R9H
        CJNE A,R9L,DECALAG ;RECHERCHE DU DECALAGE DE JOURS ENTRE ANNEE ET REF
        LJMP FINCAL ;VERS LA FIN DU CALCUL
DECALAG: CLR C
        LCALL BCD_HEX
        MOV B,#04H
        DIV AB
        MOV A,B ;TEST SI POINTEUR D'ANNEE EST BISSEC
        JNZ PASBIS
        INC REL ;+2 JOURS CHAQUE ANNEE BISS
PASBIS: INC REL ;+1 JOURS CHAQUE ANNEE NON BISS
        MOV A,R9H
        CLR C
        ADD A,#01H
        DA A
        MOV R9H,A
        LJMP DEBDEC ;REL CONTIENT EN HEXA LE JOUR DE LA SEM DU 1/01

FINCAL: CALL QANTIEM ;CALCUL DU QUANTIEME DE L'ANNEE EN COURS
        MOV A,RFL
        CLR C
        ADD A,#99H
        DA A
        JC DECPASC
        DEC RFH
DECPASC:MOV RFL,A ;DECREMENTE LE QUANTIEME

        MOV A,RFH ;RECUP CENTAINES
        RL A ;* 2
        CLR C
        ADD A,REL ;AJOUTE AU RESULTAT DE JOUR
        MOV REL,A

        MOV A,RFL
        LCALL BCD_HEX
        ADD A,REL
        MOV B,#07H
        DIV AB
        MOV A,B ;RECUP LE RESTE
        RET ;retour avec Acc=Js de 0 a 6

;** SP DE CALCUL DU QUANTIEME ANNUEL: EN SORTIE RF CONTIENT LE Q EN BCD
;** en entree R8H=le mois
;** R8L=le jour du mois
;** R9L=annee en cours

QANTIEM:PUSH R8H ;SAUVE LE MOIS
        MOV RFH,#00H
        MOV RFL,#00H ;RESULTAT DU QUANTIEME ANNUEL
        MOV RFL,R8L ;CHARGE LE JOURS DU MOIS

        MOV A,R8H ;MOIS
        CLR C
        ADD A,#99H

```

```

        DA A                                ;MOIS=MOIS-1
        MOV R8H,A
        JZ SORESUL                          ;SORTIE AVEC LE RESULTAT
MOIINF: MOV A,R8H
        MOV DPTR,#TABJOUR
        MOVC A,@A+DPTR                      ;RECUP NBR DE JOUR DS LE MOIS POINTE
        CLR C
        ADD A,RFL
        DA A
        MOV RFL,A                          ;SAUVE LE RESULTAT
        JNC PASCARY
        INC RFH                            ;INCREMENTE LES CENTAINES
PASCARY:MOV A,R8H                          ;MOIS
        CLR C
        ADD A,#99H
        DA A                                ;MOIS=MOIS-1
        MOV R8H,A
        JNZ MOIINF
        LJMP SUITEQ
;*****
;* TABLE CONTENANT LE NOMBRE DE JOUR DES MOIS
TABJOUR:DB 00H
        DB 31H,28H,31H,30H,31H,30H,31H,31H,30H
        DB 00H,00H,00H,00H,00H,00H
        DB 31H,30H,31H
;*****
SUITEQ: POP R8H                            ;RECUP LE MOIS
        MOV A,R8H                          ;MOIS
        CJNE A,#02H,ANBISS                 ;CORRECTION DU QUANTIEME SI ANNEE BISS
        LJMP RESULT
ANBISS: JC RESULT                          ;MOIS EST INFERIEUR A FEVRIER
        MOV A,R9L                          ;ANNEE EN COURS
        LCALL BCD_HEX
        MOV B,#04H
        DIV AB
        MOV A,B
        JNZ RESULT
        MOV A,RFL
        CLR C
        ADD A,#01H
        DA A
        JNC PASCENT
        INC RFH
PASCENT:MOV RFL,A
RESULT: JMP SAUVEQT                        ;RETOUR AVEC RF=QUANTIEME EN BCD
SORESUL:POP R8H
SAUVEQT:MOV DPTR,#CAS_SEC + 8              ;stockage du quantieme
        MOV A,RFH
        MOVX @DPTR,A
        INC DPL
        MOV A,RFL
        MOVX @DPTR,A
        RET

;//////////////////////////////////////////
;memorisation en 0050H de l'heure en ASCII
;    JM,MOIS,ANNEE,HEUR,MIN,JS

PREP_H: PUSH B
        PUSH 3
        PUSH 0
        MOV R3,#0
        MOV P2,R3
        MOV R0,#50H

        MOV DPTR,#CAS_SEC +3              ;jour du mois
        CALL PLACE
        MOV DPTR,#CAS_SEC +4              ;mois
        CALL PLACE
        MOV DPTR,#CAS_SEC +6              ;SIECLES
        CALL PLACE
        DEC R0
        MOV DPTR,#CAS_SEC +7              ;ANNEES
        CALL PLACE
        MOV DPTR,#CAS_SEC +2              ;HEURES
        CALL PLACE
        MOV DPTR,#CAS_SEC +1              ;MINUTES
        CALL PLACE
        MOV DPTR,#CAS_SEC +0AH            ;jour de la semaine de 0 a 6
        MOVX A,@DPTR

```

```

        MOV B,#3
        MUL AB
        MOV REL,A
        MOV B,#3

RELI:   MOV A,REL
        MOV DPTR,#TABJOU
        MOVC A,@A+DPTR
        MOVX @R0,A
        INC R0
        INC REL
        DJNZ B,RELI
        CLR A
        MOVX @R0,A

        POP 0
        POP 3
        MOV P2,R3
        POP B
        RET

PLACE:  MOVX A,@DPTR
        CALL HEX_ASC
        MOV A,REH
        MOVX @R0,A
        INC R0
        MOV A,REL
        MOVX @R0,A
        INC R0
        CLR A
        MOVX @R0,A
        INC R0
        RET

;/////////////////////////////////////:
;  emission de l'heure en affichage general sur les CP32
;  appele toute les minutes

VISU_H: JB KP_ADRE,NOAFFH ;si adresssing Kpads, les claviers ne sont plus interv
        MOV B,#DateTime
        CALL DemandeDiffusion
NOAFFH: LJMP RSP

;/////////////////////////////////////:
;////////// calcul du jour de la semaine et du quantieme a l'init
INIHEU: CALL JSEMAI
        MOV DPTR,#CAS_SEC +0AH
        MOVX @DPTR,A ;jsem de 0 a 6
        LJMP RSP

        END

```